

Corrigé de la série d'exercices N° 03

**Les fonctions Mathématiques
SOMMEPROD()**

Nadjib Benaouda

1 Rappel

1.1 Adressage relatif, absolu et mixte

La majorité des formules qu'on crée avec Excel incluent des références à des cellules ou de plage de cellules. Ces références nous aident à développer des formules qui s'adaptent dynamiquement au contenu de ces cellules (ou plage de cellules). Par exemple, si notre formule fait référence à la cellule A1 et on change la valeur contenue dans cette cellule, le résultat de la formule sera changé suivant cette nouvelle valeur. Trois types de références peuvent être utilisés:

1. **Référence relative:** lors de la recopie d'une formule dans une autre cellule, les références des cellules dans la nouvelle formule sont adaptées de manière à garder les mêmes écarts entre les formules et les cellules adressées.
2. **Référence absolue:** avec une telle référence et lors de la recopie, les références des lignes et des colonnes dans la formule qui est recopiée ne connaissent aucun changement car la référence fait référence à une cellule bien spécifique. Une référence absolue utilise deux signes dollar, un premier pour la lettre de la colonne et un deuxième pour le numéro de la ligne (e.g. \$A\$1).
3. **Référence mixte (semi-relative):** avec une telle référence, soit la référence de la ligne ou celle de la colonne est relative et l'autre est absolue. Avec une référence à ligne absolue, seule la ligne est figée et la colonne reste relative au nouvel emplacement. Une telle référence s'écrit de la façon suivante: A\$1. En contre partie, avec une référence à colonne absolue seule la colonne est figée et la ligne est relative au nouvel emplacement. Dans ce dernier cas, la référence s'écrit comme suit: \$A1.

Excel utilise par défaut l'adressage relatif. La figure (1) illustre un exemple d'un tel adressage. La formule dans la cellule F2 qui calcule la vente (montant) du vendeur Mohamed après avoir vendu 5 frigidaires de marque LG, de type No frost et dont le prix de chacun est égale 52000 est comme suit:

	A	B	C	D	E	F	G
1	Marque	Type	Vendeur	Nombre	Prix_Unit	Vente (montant)	
2	LG	No frost	Mohamed	5	52000	260000	=D2*E2
3	LG	Normal	Kamel	4	42000	168000	=D3*E3
4	LG	No frost	Ali	7	54000	378000	=D4*E4
5							

Figure 1: La recopie d'une formule contenant des références de cellules relatives.

=D2*E2

Cette formule utilise des références relatives. Donc, si on la recopie dans la cellule qui est au-dessous (cellule F3), les références s'ajustent et la formule devient:

=D3*E3

Maintenant, si la formule dans la cellule F2 utilise des références absolues, c-à-d. elle est de la forme suivante:

=D\$2*\$E\$2

Et on la recopie dans la cellule F3, la formule reste identique et le résultat sera éronné car les

références dans la formule recopiée font référence aux cellules D2 et E2 et pas aux cellules D3 et E3, cela n'est pas ce que nous souhaitons.

Imaginez-vous maintenant que les trois vendeurs (Mohamed, Kamel et Ali) et comme illustre la figure (2) ont vendu le même nombre de frigidaires (information contenue dans la cellule B6). Si on veut calculer la vente de chacun d'eux, la formule dans la cellule E2 sera comme suit:

=B\$6*E2

Cette fois ci, lorsqu'on recopie cette formule dans la cellule E3, elle devient:

=B\$6*E3

comme on peut constater, la référence à la cellule E2 a été ajustée mais pas la référence à la cellule B6, et cela est exactement ce que nous voulons.

	A	B	C	D	E	F
1	Marque	Type	Vendeur	Prix_Unit	Vente (montant)	
2	LG	No frost	Mohamed	52000	260000	=B\$6*E2
3	LG	Normal	Kamel	42000	210000	=B\$6*E3
4	LG	No frost	Ali	54000	270000	=B\$6*E4
5						
6	Nombre		5			
7						

Figure 2: La recopie d'une formule contenant une référence de cellule absolue et une deuxième relative.

la figure (3) illustre un exemple qui montre l'utilisation des références mixtes. Les formules contenues dans la plage de cellules C4:F8 calculent les aires pour différentes longueurs et largeurs. Par exemple, la formule contenue dans la cellule C4 est comme suit:

$=\$B4*C\3

Les deux références de cellules utilisées sont mixtes. La référence à la cellule B4 utilise une référence absolue de colonne (\$B) et la référence à la cellule C3 utilise une référence absolue de ligne (\$3). Comme résultat, cette formule peut être recopiée vers le bas et vers la droite et tous les calculs seront justes. À titre d'exemple, la formule dans la cellule E7 sera comme suit:

$=\$B7*E\3

Si les références dans la formule de la cellule C4 sont relatives ou absolues, les calculs seront faux.

	A	B	C	D	E	F
1						
2						
3			1	2	3	4
4		1.5	1.5	3	4.5	6
5		2	2	4	6	8
6		2.5	2.5	5	7.5	10
7		3	3	6	9	12
8		3.5	3.5	7	10.5	14
9						

Figure 3: Références Mixtes.

1.2 Fonction SOMMEPROD()

La fonction **SOMMEPROD** multiplie les valeurs correspondantes des matrices spécifiées et calcule la somme de ces produits. Sa syntaxe est comme suit:

$=\text{SOMMEPROD}(\text{matrice1}; [\text{matrice2}]; [\text{matrice3}]; \dots)$

Ses arguments sont définis comme suit:

- **matrice1**: obligatoire. Représente le premier argument de matrice dont vous voulez multiplier les valeurs pour ensuite additionner leur produit.
- **matrice2, matrice3, ...** : facultatif. Argument de matrices 2 à 255 dont vous voulez multiplier les valeurs pour ensuite additionner leur produit.

Les trois remarques suivantes sont à considérer lors de l'utilisation de la fonction SOMMEPROD:

1. Les matrices spécifiées comme arguments doivent avoir la même dimension. Si tel n'est pas le cas, SOMMEPROD renvoie la valeur #VALEUR!
2. SOMMEPROD affecte aux entrées de matrice non numérique la valeur zéro.
3. Si la fonction SOMMEPROD prend une seule matrice comme argument, elle additionne les valeurs de cette matrice et renvoie le résultat.

La figure (4) illustre un exemple d'utilisation de SOMMEPROD. La formule dans la cellule B8 multiplie toutes les valeurs des deux matrices (B5:C7 et E5:F7) puis additionne les produits. En d'autres termes, elle fait l'opération suivante: $1*2+4*8+2*3+6*4+5*1+7*7$ (118).

	A	B	C	D	E	F
1						
2						
3						
4						
5						
6						
7						
8						
9						

Figure 4: la fonction SOMMEPROD.

1.3 Nommage des cellules

Avec Excel, on peut nommer des cellules et des plages de cellules et utiliser ces noms dans les formules. L'intérêt ici est de faciliter la compréhension et la gestion des formules.

	A	B	C	D	E	F	G	H	I	J	K
1		Histoire	Géographie	Maths	Anglais						
2	Coefficient	2	2	4	3						
3						Moyenne					
4	Ali	8	5	8	11	8.27	<=	=SOMMEPROD(B\$2:E\$2;B4:E4)/SOMMEPROD(B\$2:E\$2)			
5	Omar	10	15	11	9	11.00	<=	=SOMMEPROD(B\$2:E\$2;B5:E5)/SOMMEPROD(B\$2:E\$2)			
6	Mohamed	12	18	13	16	14.55	<=	=SOMMEPROD(B\$2:E\$2;B6:E6)/SOMMEPROD(B\$2:E\$2)			
7	Said	14	19	15	16	15.82	<=	=SOMMEPROD(B\$2:E\$2;B7:E7)/SOMMEPROD(B\$2:E\$2)			
8	Nadjib	6	16	14	5	10.45	<=	=SOMMEPROD(B\$2:E\$2;B8:E8)/SOMMEPROD(B\$2:E\$2)			
9	Samir	9	3	18	12	12.00	<=	=SOMMEPROD(B\$2:E\$2;B9:E9)/SOMMEPROD(B\$2:E\$2)			
10	Wassim	16	8	5	13	9.73	<=	=SOMMEPROD(B\$2:E\$2;B10:E10)/SOMMEPROD(B\$2:E\$2)			
11	Hani	6	15	13	16	12.91	<=	=SOMMEPROD(B\$2:E\$2;B11:F11)/SOMMEPROD(B\$2:E\$2)			
12											
13			Moyennes inférieures à 10			2	<=	=SOMMEPROD((F4:F11<10)*1)			
14											

Figure 5: Exercice 01.

Moyennes inférieures à 10	=SOMMEPROD([VRAI;FAUX;FAUX;FAUX;FAUX;FAUX;VRAI;FAUX]*1)
	SOMMEPROD(matrice1; [matrice2]; [matrice3]; ...)

Figure 6: Résultat de l'application d'une opération de comparaison sur la plage F4:F11.

2 Exercice 01

La figure (5) représente le corrigé de l'exercice 01. Les cellules de la plage F4:F11 comportent des formules qui calculent pour chaque étudiant sa moyenne. La formule dans la cellule F4 qui est comme suit:

=SOMMEPROD(B\$2:E\$2;B4:E4)/SOMMEPROD(B\$2:E\$2)

additionne les produits coefficient multiplié par note du module (Partie SOMMEPROD(B\$2:E\$2;B4:E4)) et divise ensuite le résultat par la somme de coefficient (partie SOMMEPROD(B\$2:E\$2)). Il ne faut pas oublier que lorsque SOMMEPROD prend comme argument une seule plage de cellule, elle additionne les valeurs de cette plage et renvoie le résultat.

Les dollars qui sont ajoutés avant chaque numéro de la ligne 2 servent à figer cette ligne lors de la recopie de la formule vers le bas afin de faire toujours référence aux coefficients contenus dans la plage B2:E2.

La cellule F13 comporte la formule qui nous renvoie le nombre d'étudiants ayant une moyenne inférieure à 10. Cette formule est comme suit:

=SOMMEPROD((F4:F11<10)*1)

Dans cette formule, il faut savoir que le fait de taper F4:F11<10 veut dire que nous sommes en train d'appliquer une opération de comparaison sur la plage F4:F11 avec l'utilisation de l'opérateur de comparaison "<". Il est pratique de penser de cette opération comme une question que nous posons sur la plage F4:F11. Dans ce cas par exemple, on pose la question suivante: "est-ce-que l'une des valeurs de la plages F4:F11 est inférieure à 10?". Comme résultat, Excel renvoie un nouveau tableau comportant des valeurs "vrai" ou "faux" suivant si la condition est satisfaite ou pas. La figure (6) montre ce que Excel affiche après la sélection de la partie F4:F11<10 de la formule et l'appui sur la touche F9 du clavier. Les différents opérateurs de comparaison qui sont utilisés dans Excel sont présentés dans le tableau (1).

Opérateur de comparaison	Signification	Exemple
= (signe égal)	Égal à	A1=B1
> (signe supérieur à)	Supérieur à	A1>B1
< (signe inférieur à)	Inférieur à	A1<B1
>= (signe supérieur ou égal à)	Supérieur ou égal à	A1>=B1
<= (signe inférieur ou égal à)	Inférieur ou égal à	A1<=B1
<> (signe différent)	Différent de	A1<>B1

Table 1: Opérateurs de comparaison dans Excel.

Maintenant pour comprendre pourquoi on ajoute la partie "*1" dans la formule, il faut mettre en tête les deux points suivants:

	A	B	C
1	Valeur	Conversion	Formule
2	VRAI	1	=A2*1
3	FAUX	0	=A3*1
4			

(a)

Moyennes inférieures à 10	=SOMMEPROD([1;0;0;0;0;0;1;0])
	SOMMEPROD(matrice1; [matrice2]; [n

(b)

Figure 7: La conversion de VRAI en 1 et FAUX en 0 via l'utilisation de l'opération arithmétique " $\times 1$ ".

1. SOMMEPROD ne peut pas comprendre les valeurs "vrai" et "faux". Comme on a déjà mentionné, SOMMEPROD interprète les valeurs non-numériques comme des zéros.
2. Comme montre la figure (7a), l'opération arithmétique " $\times 1$ " convertit la valeur "vrai" en 1 et la valeur "faux" en 0.

Si on entre la formule sans l'ajout de l'opération " $\times 1$ ", Excel renvoie la valeur 0, car, SOMMEPROD traite "vrai" et "faux" comme des zéros. L'ajout de " $\times 1$ " et comme montre la figure (7b) permet leur conversion afin que SOMMEPROD peut additionner les valeurs numériques (0 et 1) qui résultent. Le résultat de cette addition est le nombre d'étudiants qui ont une moyenne inférieure à 10.

3 Exercice 02

Afin d'élaborer correctement les formules qui répondent aux différentes questions demandées, il est impérativement nécessaire de comprendre la table Excel qui est donnée et la sémantique de chaque question. Pour cela, je recommande vivement de tirer de la question:

1. L'opération souhaitée, c.à.d. est-ce-que nous avons à faire à un dénombrement (le nombre de vente, le nombre de frigidaire, ...etc), nous nous intéressons à la vente (montant), ...etc,
2. Les conditions ou les critères demandées,
3. La relation entre les différentes conditions (C.à.d. Savoir quelles conditions doivent être satisfaites en même temps, est-ce-que la satisfaction de l'une des conditions est suffisant, etc.). Cela permet de définir la façon à mettre les opérateurs de multiplication ($\times$) et d'addition ($+$) entre les différentes conditions.

Dans cet exercice et suivant la table de l'état de vente donnée, le nommage suivant est considéré:

Le nom	correspond à la plage
marque	A2:A31
type	B2:B31
vendeur	C2:C31
nombre	D2:D31
prix	E2:E31

Dans ce qui suit, pour chaque question et avec l'utilisation de la fonction SOMMEPROD, deux formes de formules sont données. Une première sans l'utilisation du nommage et une deuxième avec l'utilisation du nommage.

Question 01: La vente totale du vendeur mohamed. (2069000)

=SOMMEPROD((C2:C31="Mohamed")*D2:D31*E2:E31)

=SOMMEPROD((vendeur="Mohamed")*nombre*prix)

D'après cette question, on s'intéresse à la vente totale, donc le montant total, qui est récupéré par le vendeur Mohamed suivant ses différentes ventes. Donc, les ventes sommées seront seulement celles qui correspondent au vendeur mohamed.

Il faut savoir que la partie " $\times 1$ " n'est pas obligatoire ici, car, l'implication de l'une des plages *nombre* ou *prix* enlève cette obligation. Cependant, l'ajout de cette partie ne change pas le résultat renvoyé par cette formule et cette dernière reste toujours correcte.

Question 02: La vente des frigidaires no frost du vendeur Mohamed. (975000)

=SOMMEPROD((B2:B31="No frost")*(C2:C31="Mohamed")*D2:D31*E2:E31)

=SOMMEPROD((type="No frost")*(vendeur="Mohamed")*nombre*prix)

Cette fois-ci les ventes sommées sont seulement celles des frigidaires no frost du vendeur Mohamed. Donc, les deux conditions doivent être satisfaites en même temps. Pour cela, on utilise une multiplication (*) entre les deux conditions pour dire que le type de frigidaire doit être de type no frost et le vendeur doit être mohamed.

Question 03: La vente des frigidaires de type Normal du vendeur Mohamed. (1094000)

=SOMMEPROD((B2:B31="Normal")*(C2:C31="Mohamed")*D2:D31*E2:E31)

=SOMMEPROD((type="Normal")*(vendeur="Mohamed")*nombre*prix)

Question 04: La vente totale des frigidaires No frost. (7518000)

=SOMMEPROD((B2:B31="No frost")*D2:D31*E2:E31)

=SOMMEPROD((type="No frost")*nombre*prix)

Question 05: La vente totale des frigidaires de type normal. (4499000)

=SOMMEPROD((B2:B31="normal")*D2:D31*E2:E31)

=SOMMEPROD((type="normal")*nombre*prix)

Question 06: Le nombre de vente de frigidaires No frost. (18)

=SOMMEPROD((B2:B31="No frost")*1)

=SOMMEPROD((type="No frost")*1)

Attention!!! Chaque ligne de la table de l'état de vente des vendeurs représente une vente. Il ne faut pas inclure la plage nombre. Cette formule considère seulement les lignes (les ventes) qui correspondent au type No frost.

Question 07: Le nombre de ventes de frigidaires de type Normal. (12)

=SOMMEPROD((B2:B31="No frost")*1)

=SOMMEPROD((type="No frost")*1)

Question 08: Le nombre de frigidaires de type Normal vendus par mohamed. (20)

=SOMMEPROD((B2:B31="Normal")*(C2:C31="Mohamed")*D2:D31)

=SOMMEPROD((type="Normal")*(vendeur="Mohamed")*nombre)

Question 09: Le nombre de frigidaires de type Normal vendus par Mohamed et Ali. (34)

=SOMMEPROD((B2:B31="Normal")*((C2:C31="Mohamed")+(C2:C31="Ali"))*D2:D31)

=SOMMEPROD((type="Normal")*((vendeur="Mohamed")+(vendeur="Ali"))*nombre)

=SOMMEPROD(((type="Normal")*(vendeur="mohamed")+(type="Normal")*(vendeur="Ali"))*nombre)

Voici trois façons à écrire cette formule. Dans ce cas et d'après la question, le nombre des frigidaires qu'on veut savoir comporte le nombre de frigidaires de type normal vendus par Mohamed plus le nombre de frigidaires de type normal vendus par Ali. Donc, le type doit toujours être normal et le vendeur peut être Mohamed ou Ali. Comme nous avons déjà vu, on utilise la multiplication entre les conditions qui doivent être satisfaites en même temps. Mais lorsque au moins une condition doit être satisfaite on utilise l'opérateur d'addition (+).

Attention!!! n'oubliez pas de mettre les parenthèses aux endroits appropriés. En fait, Excel calcule une formule de gauche à droite suivant un certain ordre qui est bien spécifique. Cet ordre est indiqué dans le tableau (2). Dans cette série et d'après cette table, il faut seulement mettre en tête c'est que l'opération de multiplication (*) est traitée avant l'opération d'addition (+). Pour changer l'ordre de calcul, mettez entre parenthèses la partie de la formule qui doit être calculée en premier. Par exemple, la formule "2*5+3" renvoie 13 et le formule "2*(5+3)" renvoie 16.

Opérateur	Description
: (deux points)	opérateurs de référence
; (point virgule)	
(espace simple)	
-	Négation
%	Pourcentage
^	Exposant
* et /	Multiplication et division
+ et -	addition et soustraction
&	Concaténation (connecte deux chaînes de texte)
= et < >	Opérateurs de comparaison
< et >	
<= et >=	

Table 2: Ordre de priorité des opérateurs en Excel.

4 Exercice 03

Dans cet exercice et suivant la table Excel donnée, le nommage suivant est adopté:

Nom	Correspond à la plage
nom	A2:A31
mois	B2:B31
montants	C2:C31

Question 01: Le nombre de ventes d'Ali au mois de Mars. (3)

=SOMMEPROD((A2:A31="Ali")*(C2:C31="Mars"))

=SOMMEPROD((nom="Ali")*(mois="Mars"))

Question 02: La vente d'Ali au mois de Mars. (2230)

=SOMMEPROD((A2:A31="Ali")*(C2:C31="Mars")*C2:C31)

=SOMMEPROD((nom="Ali")*(mois="Mars")*montants)

Question 03: Le nombre de ventes de Mohamed. (8)

=SOMMEPROD((A2:A31="Mohamed")*1)

=SOMMEPROD((nom="Mohamed")*1)

Question 04: Le nombre de ventes supérieures à 600 DZD au mois de Janvier. (2)

=SOMMEPROD((C2:C31>600)*(B2:B31="Janvier"))

=SOMMEPROD((montants>600)*(mois="Janvier"))

Question 05: La vente d'Omar en Janvier et Mars. (2760)

=SOMMEPROD((A2:A31="Omar")*((B2:B31="Janvier")+(B2:B31="Mars"))*C2:C31)

=SOMMEPROD((nom="Omar")*((mois="Janvier")+(mois="Mars"))*montants)

Attention !!! n'oubliez pas les parenthèses. Si on écrit cette formule de la façon suivante:

=SOMMEPROD((nom="Omar")*(mois="Janvier")+(mois="Mars")*montants)

Alors le résultat renvoyé sera 7043, ce qui est faux.

Question 06: Le nombre de ventes d'Ahmed et Mohamed au mois de Janvier et Mars. (9)

=SOMMEPROD(((A2:A31="Ahmed")+(A2:A31="Mohamed"))*((B2:B31="Janvier")+(B2:B31="Mars")))

=SOMMEPROD(((nom="Ahmed")+(nom="Mohamed"))*((mois="Janvier")+(mois="Mars")))

Question 07: La vente d'Ahmed et Mohamed au mois de Janvier et Mars. (4740)

=SOMMEPROD(((A2:A31="Ahmed")+(A2:A31="Mohamed"))*((B2:B31="Janvier")+(B2:B31="Mars"))*montants)

=SOMMEPROD(((nom="Ahmed")+(nom="Mohamed"))*((mois="Janvier")+(mois="Mars"))*montants)